

NAG Library Routine Document

G05TBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05TBF generates a vector of pseudorandom logical values – .TRUE. with probability p and .FALSE. with probability $(1 - p)$.

2 Specification

```
SUBROUTINE G05TBF(N, P, STATE, X, IFAIL)
INTEGER          N, STATE(lstate), IFAIL
double precision P
LOGICAL          X(N)
```

3 Description

G05TBF generates n logical values x_i from the relation

$$y_i < p$$

where y_i is a pseudorandom number from a uniform distribution over $(0, 1]$, generated by G05SAF using the values of STATE as input to this routine.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05TBF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the number of pseudorandom logical values to be generated.
Constraint: $N \geq 0$.
- 2: P – **double precision** *Input*
On entry: must contain the probability of G05TBF returning .TRUE..
Constraint: $0.0 \leq P \leq 1.0$.
- 3: STATE(lstate) – INTEGER array *Communication Array*
Note: lstate = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 4: X(N) – LOGICAL array *Output*
On exit: the n logical values.

5: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 2

On entry, $P < 0$,
or $P > 1$.

IFAIL = 3

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints the first 20 pseudorandom logical values generated by G05TBF after initialization by G05KFF, when the probability of a .TRUE. value is 0.5.

9.1 Program Text

```
*      G05TBF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSEED, MSTATE, N
      PARAMETER        (MSEED=1,MSTATE=633,N=20)
*      .. Local Scalars ..
      DOUBLE PRECISION P
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
      INTEGER          SEED(MSEED), STATE(MSTATE)
      LOGICAL          X(N)
*      .. External Subroutines ..
      EXTERNAL         G05KFF, G05TBF
*      .. Executable Statements ..
```

```

        WRITE (NOUT,*) 'G05TBF Example Program Results'
        WRITE (NOUT,*)
*       Set the distribution parameters P
        P = 0.5D0
*       Initialize the seed
        SEED(1) = 1762543
*       GENID and SUBID identify the base generator
        GENID = 1
        SUBID = 1
*       Initialize the generator to a repeatable sequence
        LSTATE = MSTATE
        LSEED = MSEED
        IFAIL = 1
        CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
        IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99998) IFAIL
            GO TO 20
        END IF
*       Generate the variates
        IFAIL = 1
        CALL G05TBF(N,P,STATE,X,IFAIL)
        IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99997) IFAIL
            GO TO 20
        END IF
*       Display the variates
        WRITE (NOUT,99999) (X(I),I=1,N)
*
    20 CONTINUE
*
99999 FORMAT (1X,L1)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05TBF returned with IFAIL = ',I5)
        END

```

9.2 Program Data

None.

9.3 Program Results

G05TBF Example Program Results

```

F
T
F
F
T
T
T
F
T
F
T
T
F
T
F
T
F
F
F

```
